# Journal of Software for Algebra and Geometry


## Tropical computations for toric intersection theory in Macaulay2


ALESSIO BORZÌ



vol 14 2024

JSAG **14** (2024), 19–29
https://doi.org/10.2140/jsag.2024.14.19

The Journal of Software for
Algebra and Geometry

# Tropical computations for toric intersection theory in Macaulay2

ALESSIO BORZÌ

ABSTRACT: We present the Macaulay2 package `TropicalToric` for toric intersection theory computations using tropical geometry.

**1.** INTRODUCTION. Toric varieties are ubiquitous in algebraic geometry. Their intersection theory was first studied in Fulton and Sturmfels [12], and it has many applications in different contexts, including wonderful and tropical compactifications [7; 10; 34], birational geometry [5; 14; 15], tropical geometry [23], tropical intersection theory [2; 21; 22; 29; 30] and combinatorial Hodge theory [1; 19; 20].

In a certain way, intersection classes of a toric variety with fan $\Sigma$ can be thought of in terms of balanced subfans of $\Sigma$, also referred to as *Minkowski weights*; see [12] or [27, Theorem 6.7.5]. From the structure theorem of tropical geometry [27, Theorem 3.3.5], we know that the tropicalization of a subvariety of a torus $Y \subseteq T^n$ is a balanced fan. A surprising connection between tropical and toric geometry is that the tropicalization of $Y$ is the balanced fan corresponding to the intersection class of the closure of $Y$ inside an "enough refined" toric variety; see Theorem 3.1 for a more precise statement. This fact allows us to compute toric intersection classes starting from the data of the tropicalization.

We present a new package, `TropicalToric`, for Macaulay2 [26]. The package implements toric cycles and intersection products on simplicial toric varieties (Section 2), and, following the ideas outlined above, allows us to compute the intersection class of an irreducible subvariety of a simplicial toric variety not contained in the toric boundary, from the data of its tropicalization (Section 3). The tropicalization is performed with the use of the Macaulay2 package `Tropical` [3]. Further, we present some applications to the intersection theory of wonderful compactifications and the moduli space $\overline{M}_{0,n}$ and illustrate an example in a multiprojective space using a theorem of Huh and Katz [20] about characteristic polynomials of realizable matroids.

**2.** TORIC INTERSECTION THEORY. In this section, we review the basics of toric intersection theory, for more information see [12], [6, Section 12.5] or [27, Section 6.7]. In addition, we showcase how it is implemented in the package.

Let $X_\Sigma$ be a smooth complete toric variety of dimension $n$ with fan $\Sigma$. We denote by $\Sigma(k)$ the cones of $\Sigma$ of dimension $k$, with $Z^k(X_\Sigma) = Z_{n-k}(X_\Sigma)$ the group of codimension $k$ cycles and with $A^k(X_\Sigma) = A_{n-k}(X_\Sigma)$ the codimension $k$ Chow group, that is, the group of codimension $k$ cycles modulo

*MSC2020:* 14Q99, 14T20.
*Keywords:* tropical geometry, toric geometry, intersection theory, wonderful compactifications.
`TropicalToric` version 1.0

rational equivalence. The codimension $k$ Chow group $A^k(X_\Sigma)$ is generated by the set $\{[V(\sigma)] : \sigma \in \Sigma(k)\}$ of classes of orbit closures of codimension $k$, and the relations in each Chow group can be described in an explicit way; see [12, Proposition 2.1].

We can show (see [11, Chapter 8]) that there is an intersection product $A^k(X_\Sigma) \times A^r(X_\Sigma) \to A^{k+r}(X_\Sigma)$ that makes $A^*(X_\Sigma) = \bigoplus_{k=0}^n A^k(X_\Sigma)$ into a graded ring, called the *Chow ring* of $X_\Sigma$. If we now assume that $X_\Sigma$ is just complete and simplicial, then the intersection product can be defined on rational cycles, making $A^*(X_\Sigma)_\mathbb{Q} = A^*(X_\Sigma) \otimes \mathbb{Q}$ into a graded ring. The structure of the Chow ring has an explicit description; see, for instance, [27, Theorem 6.7.1].

Our Macaulay2 package implements toric cycles and the intersection product as in [6, Lemma 12.5.2].

**Example 2.1.** Let $X_\Sigma$ be the blow-up of $\mathbb{P}^2$ at one of the coordinate points, where the fan $\Sigma$ and the primitive ray vectors of its rays are




$$\begin{array}{c} \begin{array}{cccc} \rho_0 & \rho_1 & \rho_2 & \rho_3 \end{array} \\ \begin{bmatrix} 1 & 1 & 0 & -1 \\ 0 & 1 & 1 & -1 \end{bmatrix} \end{array}$$

Let $H$ be the strict transform in $X_\Sigma$ of a general line in $\mathbb{P}^2$ and $E$ be the exceptional divisor. The Picard group of $X_\Sigma$ is generated by the classes of these two divisors $\mathrm{Pic}(X_\Sigma) = \langle [H], [E] \rangle$. With the notation above, we have

$$[V(\rho_0)] = [H] - [E], \quad [V(\rho_1)] = [E], \quad [V(\rho_2)] = [H] - [E], \quad [V(\rho_3)] = [H].$$

Now we verify with our package that the divisor class $[V(\rho_1)]$ has a negative self-intersection.

```
i1 : needsPackage "TropicalToric";

i2 : raysList = {{1,0},{1,1},{0,1},{-1,-1}};

i3 : coneList = {{0,1},{1,2},{2,3},{3,0}};

i4 : X = normalToricVariety (raysList, coneList);
```

Now define the toric cycle $V(\rho_1)$.

```
i5 : E = X_{1}

o5 = X
      {1}

o5 : ToricCycle on X
```

We point out that the type `ToricCycle` should not be confused with the type `ToricDivisor` from the `NormalToricVarieties` package. The toric cycle $V(\sigma)$ of the normal toric variety `X` associated to the cone $\sigma$ given by a list of rays `L` is defined with the command `X_L`. For example, `X_{1,2}` or `X_{0}` define toric cycles, whereas `X_1` defines a toric divisor. We are allowed only to multiply a toric cycle with a toric divisor. Now, we finally compute the self intersection of $E$:

```
i6 : X_1 * E

o6 = - X
        {1, 2}

o6 : ToricCycle on X
```

The resulting cycle $-V(\rho_1+\rho_2)$ is rationally equivalent to $E^2$. The negative sign tells us that the self-intersection number of the exceptional divisor is $-1$. We can compute the degree of maximal codimension cycles with `degCycle`:

```
i7 : degCycle(-X_{1,2})
o7 = -1
```

## 3. Tropical computations.

In this section, we describe and showcase the algorithm implemented in the main function of the package `classFromTropical` that computes the intersection class of an irreducible subvariety of a smooth toric variety from its tropicalization.

The algorithm is mainly based on the following result, that appears in various versions in the literature; see, for instance, [23, Lemma 2.3], [21, Section 9] or [27, Theorem 6.7.7].

**Theorem 3.1.** *Let $Y$ be a subvariety of the algebraic torus $T^n$, and let $\overline{Y}$ be its closure in a toric variety $X_\Sigma$ such that $|\Sigma| = \operatorname{trop}(Y)$ and $\Sigma$ is simplicial. Let $\Sigma'$ be a simplicial completion of the fan $\Sigma$, and let $i : X_\Sigma \to X_{\Sigma'}$ be the induced inclusion. Then, for every maximal cone $\sigma$ in $\Sigma$, we have*

$$m(\sigma) = \deg\bigl([i_*(\overline{Y})]\cdot[V(\sigma)]\bigr),$$

*where $m(\sigma)$ is the multiplicity of $\sigma$ in* $\operatorname{trop}(Y)$.

Now let $Y$ be an irreducible $k$-dimensional subvariety of an $n$-dimensional simplicial toric variety $X_\Sigma$, and suppose that $Y \cap T^n \neq \varnothing$. Note that in this setting we cannot directly apply Theorem 3.1 since $\operatorname{trop}(Y)$ is not necessarily a subfan of $\Sigma$.

In order to compute the class of $Y$ in the Chow ring of $X_\Sigma$, we proceed as follows. First, let $\Sigma'$ be a completion of $\Sigma$ and $i : X_\Sigma \to X_{\Sigma'}$ be the induced inclusion. Now let $\widetilde{\Sigma}$ be a refinement of $\Sigma'$ such that it contains a subfan with support the tropicalization of $Y \cap T^n$, and let $\pi : X_{\widetilde{\Sigma}} \to X_{\Sigma'}$ be the induced toric map. From [12, Proposition 2.4], we have an isomorphism $A_k(X_{\Sigma'}) \simeq \operatorname{Hom}(A^k(X_{\Sigma'}), \mathbb{Z})$ mapping a class $[Z]$ to the homomorphism $[Z'] \mapsto \deg([Z]\cdot[Z'])$. Therefore, in order to compute the class $[Y] \in A_k(X_\Sigma)$, it is enough to compute the intersection numbers $\deg([i_*(Y)]\cdot[V(\sigma)])$ for every $\sigma \in \Sigma'(k)$, as the classes $[V(\sigma)]$ generate $A^k(X_{\Sigma'})$. Let $Y'$ be the strict transform of $Y$ in $X_{\widetilde{\Sigma}}$. From the projection formula [11, Proposition 2.3 (c)], we have

$$[i_*(Y)]\cdot[V(\sigma)] = \pi_*([Y'])\cdot[V(\sigma)] = \pi_*\bigl([Y']\cdot\pi^*([V(\sigma)])\bigr),$$

from which it follows that $\deg\bigl([i_*(Y)]\cdot[V(\sigma)]\bigr) = \deg\bigl([Y']\cdot\pi^*([V(\sigma)])\bigr)$. These last intersection numbers can be computed from the tropicalization of $Y \cap T^n$ by using Theorem 3.1, since $\deg\bigl([Y']\cdot[V(\sigma')]\bigr)$ is the multiplicity of the cone $\sigma' \in \widetilde{\Sigma}(k)$ in the tropicalization of $Y \cap T^n$.

The algorithm described above, while working on any simplicial toric variety, requires to compute a completion. This can be avoided by requiring the toric variety $X_\Sigma$ to be smooth. In fact, the only step in which we are really using the completion is when we apply [12, Proposition 2.4] (sometimes called *Kronecker duality*). If the variety $X_\Sigma$ is smooth, this can be substituted instead with Poincaré duality.

The function classFromTropical performs the above algorithm to compute a toric cycle rationally equivalent to a given irreducible subvariety $Y$ of a smooth toric variety $X_\Sigma$ (by using Poincaré duality). The input of the function consists of the toric variety $X_\Sigma$ and the ideal $I$ of $Y \cap T^n$ of the Laurent ring of $T^n$. Since Laurent rings are not implemented in Macaulay2, the actual input will be instead the saturation of $I$ with respect to the product of the variables in the polynomial ring:

```
i2 : X = toricProjectiveSpace 2;

i3 : R = QQ[x,y];

i4 : I = ideal(x+y+1);

i5 : classFromTropical(X,I)

o5 = X
      {0}

o5 : ToricCycle on X

i6 : J = ideal(x*y + x + y);

i7 : classFromTropical(X,J)

o7 = 2*X
        {0}

o7 : ToricCycle on X
```

The function classFromTropicalCox allows us to input the ideal of $Y$ in the Cox ring of $X_\Sigma$:

```
i8 : R = ring X;

i9 : I = ideal(R_0+R_1+R_2);

i10 : classFromTropicalCox(X,I)

o10 = X
       {0}

o10 : ToricCycle on X
```

## 4. Applications.

**4A.** ***Wonderful compactifications.*** Let $\mathcal{A}$ be an essential hyperplane arrangement of $n+1$ hyperplanes in $\mathbb{P}^d$. The intersection lattice $\mathcal{L}(\mathcal{A})$ of $\mathcal{A}$ is isomorphic to the lattice of flats of the underlying matroid $M$ of $\mathcal{A}$ [32, Proposition 3.6]. Fix a building set $\mathcal{G}$ of the lattice of flats of $M$ (see [10, Section 2]), let $\Sigma \subseteq \mathbb{R}^{n+1}/\mathbb{R}\mathbf{1} \simeq \mathbb{R}^n$ be the Bergman fan of $M$ with respect to $\mathcal{G}$ (see [27, Chapter 4]) and let $X_\Sigma$ be its associated toric variety. From [27, Proposition 4.1.1], the hyperplane arrangement complement $Y = \mathbb{P}^d \setminus \cup\mathcal{A}$ is naturally isomorphic to a linear subspace of the algebraic torus $T^n$. Thus we can embed $Y$ inside the toric variety $X_\Sigma$ and consider its closure $\overline{Y}$. This compactification coincides with the so-called De Concini–Procesi *wonderful compactification* [7], with respect to the building set $\mathcal{G}$ (see [34, Section 4]). The next result follows from [7, Theorem 3.2]; see also [9, Definition 2.3].

**Proposition 4.1.** *Let $X_1, \ldots, X_t$ be a linear extension of the opposite order of $\mathcal{L}(\mathcal{A})$. The wonderful compactification $\overline{Y}$ is the result of successively blowing up $\mathbb{P}^d$ at (the strict transforms of) $X_1, \ldots, X_t$.*

In [10], Feichtner and Yuzvinsky showed that the cohomology of $\overline{Y}$ agrees with that of $X_\Sigma$. Since both varieties are *homology isomorphism schemes* (in the sense of the definition in the appendix of [24]), their Chow rings coincide as well.

**Theorem 4.2** [27, Theorem 6.7.14]. *Let $\overline{Y}$ be a wonderful compactification of a hyperplane arrangement $\mathcal{A}$ with respect to a building set $\mathcal{G}$, and let $\Sigma$ be the associated Bergman fan. Then*

$$A^*(\overline{Y}) \simeq A^*(X_\Sigma),$$

*where the above isomorphism is the pullback map induced by the inclusion.*

The above theorem allows us to view the intersection classes of a wonderful compactification as intersection classes of the associated toric variety. Thus, we can use our package to perform intersection theory computations on wonderful compactifications.

**Example 4.3.** Let $\mathcal{A}$ be a line arrangement consisting of 4 lines $L_0$, $L_1$, $L_2$, $L_3$ in $\mathbb{P}^2$ given by the equations $x_0 = 0$, $x_1 = 0$, $x_2 = 0$, $x_0 + x_1 = 0$, respectively. Let $A$ be the matrix with columns the normal vectors of the lines $L_i$, and let $P_1$, $P_2$, $P_3$, $P_4$ be the points of intersection of the lines of $\mathcal{A}$, depicted as

$$A = \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$




The underlying matroid $M$ of $\mathcal{A}$, on the ground set {0, 1, 2, 3}, is realized by the matrix $A$ by labeling the columns with 0, 1, 2, 3, respectively. The lattice of flats $\mathcal{L}(M)$ of $M$ is represented by




There are four rank 1 flats, corresponding to the lines $L_0$, $L_1$, $L_2$, $L_3$, and four rank 2 flats, corresponding to the points $P_1$, $P_2$, $P_3$, $P_4$. Let $\mathcal{G} = \mathcal{L}(M) \setminus \{\varnothing\}$ be the maximal building set of $\mathcal{L}(M)$. Then, from Proposition 4.1, the wonderful compactification $\overline{Y}$ of the complement $Y = \mathbb{P}^2 \setminus \cup\mathcal{A}$ with respect to $\mathcal{G}$ is the blow-up of $\mathbb{P}^2$ at the points $P_1$, $P_2$, $P_3$, $P_4$. In particular, $\overline{Y}$ is a smooth projective surface, all Weil divisors are Cartier [17, Proposition II.6.11] and the class group is isomorphic to the Picard group [17, Corollary II.6.16]. From [17, Proposition V.3.2], the Picard group of $\overline{Y}$ has a basis given by

$$\operatorname{Pic}(\overline{Y}) = \langle [H], [E_1], \ldots, [E_t] \rangle, \tag{1}$$

where $[H]$ is the class of the strict transform $H$ of a general line in $\mathbb{P}^2$, and $[E_i]$ is the class of the exceptional divisor $E_i$ of the blow-up at $P_i$.

The Bergman fan $\Sigma \subseteq \mathbb{R}^4/\mathbb{R}\mathbf{1}$ of $M$ with respect to $\mathcal{G}$ has eight rays, denoted $\{\rho_i : 0 \le i \le 7\}$. Their primitive ray vectors are given by the columns of the matrix

$$\begin{array}{cccccccc} \rho_0 & \rho_1 & \rho_2 & \rho_3 & \rho_4 & \rho_5 & \rho_6 & \rho_7 \end{array}$$
$$\begin{bmatrix} 1 & 0 & 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 1 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 0 & 1 & 1 & 1 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 & 1 \end{bmatrix}$$

where $\rho_0$, $\rho_1$, $\rho_2$, $\rho_3$ correspond to the rank 1 flats in $\mathcal{G}$, which in turn correspond to the lines $L_0$, $L_1$, $L_2$, $L_3$, respectively, and $\rho_4$, $\rho_5$, $\rho_6$, $\rho_7$ correspond to the rank 2 flats in $\mathcal{G}$, that correspond to the points $P_1$, $P_2$, $P_3$, $P_4$, respectively. Since $\mathcal{G}$ is the maximal building set, the maximal cones of $\Sigma$ are just the maximal chains of the lattice of flats $\mathcal{L}(M)$.

By using the isomorphism in Theorem 4.2, let $[Y_{\rho_i}]$ denote the class in $A^*(\overline{Y})$ isomorphic to the class of the torus invariant divisor of $X_\Sigma$ associated to the ray $\rho_i$. Expressing these divisors in the Picard basis (1), we have

$$\begin{aligned} [Y_{\rho_0}] &= [H]-[E_1]-[E_2], & [Y_{\rho_4}] &= [E_1], \\ [Y_{\rho_1}] &= [H]-[E_1]-[E_3], & [Y_{\rho_5}] &= [E_2], \\ [Y_{\rho_2}] &= [H]-[E_2]-[E_3]-[E_4], & [Y_{\rho_6}] &= [E_3], \\ [Y_{\rho_3}] &= [H]-[E_1]-[E_4], & [Y_{\rho_7}] &= [E_4]. \end{aligned} \tag{2}$$

Let $\mathbb{C}[y_0^{\pm 1}, y_1^{\pm 1}, y_2^{\pm 1}]$ be the Laurent ring of the torus

$$T^3 = \{(1 : y_0 : y_1 : y_2) : y_0, y_1, y_2 \in \mathbb{C}^*\} \subseteq \mathbb{P}^3.$$

The embedding $Y \hookrightarrow T^3$ is given by $(x_0 : x_1 : x_2) \mapsto (x_0 : x_1 : x_2 : x_0 + x_1)$, and the Laurent ideal of $Y$ inside $T^3$ is $I = (-1 - y_0 + y_2)$.

Now, let $C$ be the conic in $\mathbb{P}^2$ passing through the points $P_1$, $P_2$ and $P_3$ given by the equation $x_0x_1 + x_0x_2 + x_1x_2$. The ideal of $C$ in $T^3$ is $(y_0 + y_1 + y_0y_1) + I$. We expect the class of its strict transform in $\overline{Y}$ to be $[2H - E_1 - E_2 - E_3]$. We now verify this with our package, using the function `classWonderfulCompactification`.

```
i2 : R = QQ[y_0,y_1,y_2];

i3 : I = ideal(-1-y_0+y_2);

i4 : f = y_0+y_1+y_0*y_1;

i5 : raysList = {{-1,-1,-1},{1,0,0},{0,1,0},
                 {0,0,1},{0,-1,0},{-1,0,-1},
                 {1,1,0},{0,1,1}};

i6 : conesList = {{4,0},{4,1},{4,3},{5,0},{5,2},
                  {6,1},{6,2},{7,2},{7,3}};

i7 : X = normalToricVariety (raysList, conesList);

i8 : D = classWonderfulCompactification(X,I,f)

o8 = X    + X    + X
      {0}    {4}    {1}

o8 : ToricCycle on X
```

To check that this is the result we expect, compare with (2). Note that we have (tropically) dehomogenized the rays of $X_\Sigma$ with respect to the first coordinate in order to be consistent with our choice of coordinates of $T^3$.

**4B.** ***The moduli space*** $\overline{M}_{0,n}$**.** The Deligne–Mumford compactification of the moduli space $M_{0,n}$ can be realized as a wonderful compactification; see, for instance, [27, Example 6.7.16]. Therefore, we can apply to $\overline{M}_{0,n}$ the machinery described in the previous section. As an application, we compute one of the 15 Keel–Vermeire divisors of $\overline{M}_{0,6}$, using one of the equations listed in [16, Table 2]. These divisors, found independently by Keel and Vermeire [35], were the first example of an effective divisor of $\overline{M}_{0,n}$ whose class lies outside the cone generated by the classes of the boundary divisors, answering in the negative a conjecture of Fulton; see [25].

```
i2 : R = QQ[x_0..x_8];

i3 : I = ideal {-x_0+x_3+x_4, -x_1+x_3+x_5,-x_2+x_3+x_6,
               -x_0+x_2+x_7, -x_1+x_2+x_8, -x_0+x_1+1};

i4 : X = normalToricVariety fan tropicalVariety I;

i5 : f = x_0*x_1-x_2*x_3;

i6 : D = classWonderfulCompactification(X,I,f);

i7 : D = toricDivisorFromCycle(D)

o7 = X  - X  - 2*X  + X  + 2*X  + 2*X   - X   + 2*X   + 2*X   - X
      2    5      6    7      9      10    11      13      14    17

o7 : ToricDivisor on X
```

Now fix the Picard basis of $X_\Sigma$ given by the boundary divisors associated to the rays of $\Sigma$ with primitive ray vectors not equal to the standard vectors $e_i$. The complement of this Picard basis is indexed by the list $l = \{0, 1, 2, 4, 5, 7, 11, 13, 21\}$. The function `makeTransverse` computes a divisor linearly equivalent to a given divisor `D`, with support disjoint from a given list `l`. We use this function to compute a representation of the class of the Keel–Vermeire divisor computed above, in the Picard basis we fixed:

```
i8 : l = {0,1,2,4,5,7,11,13,21};

i9 : D = makeTransverse(D,l)

o9 = X  - X  - X  + X  + 2*X   - X   - X   + X   + 2*X   + 2*X
      3    6    8    9      10    14    16    17      18      19
     + 2*X   - X   - X
          20    22    24

o9 : ToricDivisor on X
```

Finally, we verify that the obtained divisor is outside the cone generated by the classes of boundary divisors. In order to do so, we interface with Polymake [13] by using the function `polymakeConeContains`:

```
i10 : D = apply(#rays X, i->D#i);

i11 : Bdivisors = apply(#rays X, i-> makeTransverse(X_i,l));

i12 : Bdivisors = apply(Bdivisors, B-> apply(#rays X, i->B#i));

i13 : polymakeConeContains(D,Bdivisors)

o13 = false
```

In [18] it was proved, by using computational methods, that the boundary divisors and the Keel–Vermeire divisors generate the effective cone of $\overline{M}_{0,6}$. In [5] it was proved that the effective cone of $\overline{M}_{0,n}$

for $n \geq 10$ is not polyhedral. The problem of determining the effective cone of $\overline{M}_{0,n}$ for $n \in \{7, 8, 9\}$ is still open. Some examples of extremal effective divisors on $\overline{M}_{0,7}$ were found in [4; 8; 28]. We performed computations similar to those displayed above on $\overline{M}_{0,7}$ and found the mentioned examples with a brute-force approach. More recently, in [31], several thousands of extremal effective divisors on $\overline{M}_{0,7}$ were found.

**4C. *Characteristic polynomials.*** Our last application is an explicit verification of a theorem proved by Huh and Katz [20] about characteristic polynomials of realizable matroids.

Let $\mathcal{A}$ be an arrangement of $n+1$ hyperplanes on $\mathbb{P}^d$, let $M$ be its underlying matroid of rank $d+1$, and let $\mathcal{L}(M)$ be the lattice of flats of $M$. The *characteristic polynomial* of $M$ is

$$\chi_M(q) = \sum_{F \in \mathcal{L}(M)} \mu(\varnothing, F) q^{d+1-r(F)},$$

where $\mu$ is the Möbius function of $\mathcal{L}(M)$; see [33, Section 3.7]. The *reduced characteristic polynomial* of $M$ is $\overline{\chi}_M(q) = \chi_M(q)/(q-1)$.

Now we embed the complement $Y = \mathbb{P}^d \setminus \cup\mathcal{A}$ in $T^n \subseteq \mathbb{P}^n$, as described in Section 4A, and consider the Cremona map

$$\varphi : \mathbb{P}^n \dashrightarrow \mathbb{P}^n, \quad (x_0, \dots, x_n) \mapsto (x_0^{-1}, \dots, x_n^{-1}).$$

Finally, let $\bar{Z}$ be the closure in $\mathbb{P}^n \times \mathbb{P}^n$ of the graph $Z$ of the restriction $\varphi_{|Y}$.

**Theorem 4.4** (Huh and Katz [20]). *Define the integers $a_i \in \mathbb{Z}$ by the formula*

$$\overline{\chi}_M(q) = \sum_{i=0}^{d} (-1)^i a_i q^{d-i}.$$

*Then*

$$[\bar{Z}] = \sum_{i=0}^{d} a_i [\mathbb{P}^{r-i} \times \mathbb{P}^i] \in A_d(\mathbb{P}^n \times \mathbb{P}^n).$$

**Example 4.5.** Let $G$ be the graph

$G =$ 

$$A = \begin{bmatrix} 1 & 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & -1 & 0 \\ 0 & 0 & 1 & 0 & -1 \end{bmatrix}.$$

Let $M$ be the rank 3 graphic matroid of $G$, realized by the matrix $A$ above. The characteristic polynomial of $M$ coincide with the chromatic polynomial of $G$. Let $\mathbb{C}[x_0^{\pm 1}, x_1^{\pm 1}, x_2^{\pm 1}, x_3^{\pm 1}]$ be the Laurent ring of the torus

$$T^4 = \{(1 : x_0 : x_1 : x_2 : x_3) : x_0, x_1, x_2, x_3 \in \mathbb{C}^*\} \subseteq \mathbb{P}^4.$$

Let $\mathcal{A}$ be the hyperplane arrangement realizing $M$. More explicitly, the normal vectors of its hyperplanes are the columns of the matrix $A$. The Laurent ideal of the hyperplane arrangement complement $Y = \mathbb{P}^2 \setminus \cup\mathcal{A}$ embedded in $T^4 \subseteq \mathbb{P}^4$ is given by $I = (-1 + x_0 + x_2, -1 + x_1 + x_3)$.

Now consider a copy of $T^4$ with Laurent ring $\mathbb{C}[x_4^{\pm 1}, x_5^{\pm 1}, x_6^{\pm 1}, x_7^{\pm 1}]$. Let $Z \subseteq T^4 \times T^4$ be the graph of the Cremona map $\varphi : \mathbb{P}^4 \dashrightarrow \mathbb{P}^4$ restricted to $Y$. The ideal of $Z$ in the Laurent ring $\mathbb{C}[x_0^{\pm 1}, \dots, x_7^{\pm 1}]$ of $T^4 \times T^4$ is generated by $I$ and the polynomials $x_i x_{i+4} - 1$ for $i \in \{0, 1, 2, 3\}$.

```
i2 : R = QQ[x_0..x_7];

i3 : I = ideal(-1+x_0+x_2,-1+x_1+x_3,
            x_0*x_4-1,x_1*x_5-1,x_2*x_6-1,x_3*x_7-1);

o3 : Ideal of R

i4 : P4 = toricProjectiveSpace 4;

i5 : X = NormalToricVarieties$cartesianProduct(P4,P4);

i6 : D = classFromTropical(X,I)

o6 = 4*X              + 4*X                + X
        {0,1,2,3,5,6}      {0,1,2,5,6,7}       {0,1,5,6,7,8}

o6 : ToricCycle on X
```

We obtained $[\bar{Z}] = [\mathbb{P}^2 \times \mathbb{P}^0] + 4[\mathbb{P}^1 \times \mathbb{P}^1] + 4[\mathbb{P}^0 \times \mathbb{P}^2]$. We now verify that the coefficients of this class are the same, up to sign, to those of the (reduced) chromatic polynomial of $G$:

```
i7 : needsPackage "Graphs";

i8 : G = graph({{0,1},{1,2},{2,3},{3,0},{0,2}});

i9 : p = chromaticPolynomial G
         4     3     2
o9 = x  - 5x  + 8x  - 4x

o9 : ZZ[x]

i10 : x = (ring p)_0;

i11 : p/(x-1)
         3     2
o11 = x  - 4x  + 4x

o11 : frac(ZZ[x])
```

ACKNOWLEDGEMENTS. I am grateful to my advisor, Diane Maclagan, for her guidance and suggestions during the development of this project. Part of the Macaulay2 package presented is based on some code previously written by Diane Maclagan, Sameera Vemulapalli, Corey Harris, Erika Pirnes and Ritvik Ramkumar.

SUPPLEMENT. The online supplement contains version 1.0 of `TropicalToric`.

REFERENCES.

[1] K. Adiprasito, J. Huh, and E. Katz, "Hodge theory for combinatorial geometries", *Ann. of Math.* (2) **188**:2 (2018), 381–452. MR Zbl

[2] L. Allermann and J. Rau, "First steps in tropical intersection theory", *Math. Z.* **264**:3 (2010), 633–670. MR Zbl

[3] C. Améndola, K. Kohn, S. Lamboglia, D. Maclagan, B. Smith, J. Sommars, P. Tripoli, and M. Zajaczkowska, "Computing tropical varieties in Macaulay2", preprint, 2017. Zbl arXiv 1710.10651

[4] A.-M. Castravet and J. Tevelev, "Hypertrees, projections, and moduli of stable rational curves", *J. Reine Angew. Math.* **675** (2013), 121–180. MR Zbl

[5] A.-M. Castravet, A. Laface, J. Tevelev, and L. Ugaglia, "Blown-up toric surfaces with non-polyhedral effective cone", *J. Reine Angew. Math.* **800** (2023), 1–44. MR Zbl

[6] D. A. Cox, J. B. Little, and H. K. Schenck, *Toric varieties*, Graduate Studies in Mathematics **124**, American Mathematical Society, Providence, RI, 2011. MR Zbl

[7] C. De Concini and C. Procesi, "Wonderful models of subspace arrangements", *Selecta Math.* (*N.S.*) **1**:3 (1995), 459–494. MR Zbl

[8] B. Doran, N. Giansiracusa, and D. Jensen, "A simplicial approach to effective divisors in $\overline{M}_{0,n}$", *Int. Math. Res. Not.* **2017**:2 (2017), 529–565. MR Zbl

[9] E. M. Feichtner, "De Concini–Procesi wonderful arrangement models: a discrete geometer's point of view", pp. 333–360 in *Combinatorial and computational geometry*, edited by J. E. Goodman et al., Math. Sci. Res. Inst. Publ. **52**, Cambridge Univ. Press, 2005. MR Zbl

[10] E. M. Feichtner and S. Yuzvinsky, "Chow rings of toric varieties defined by atomic lattices", *Invent. Math.* **155**:3 (2004), 515–536. MR Zbl

[11] W. Fulton, *Intersection theory*, 2nd ed., Ergebnisse der Math. (3) **2**, Springer, Berlin, 1998. MR Zbl

[12] W. Fulton and B. Sturmfels, "Intersection theory on toric varieties", *Topology* **36**:2 (1997), 335–353. MR Zbl

[13] E. Gawrilow and M. Joswig, "`polymake`: a framework for analyzing convex polytopes", pp. 43–73 in *Polytopes — combinatorics and computation* (Oberwolfach, 1997), edited by G. Kalai and G. M. Ziegler, DMV Sem. **29**, Birkhäuser, Basel, 2000. MR Zbl

[14] A. Gibney and D. Maclagan, "Equations for Chow and Hilbert quotients", *Algebra Number Theory* **4**:7 (2010), 855–885. MR Zbl

[15] A. Gibney and D. Maclagan, "Lower and upper bounds for nef cones", *Int. Math. Res. Not.* **2012**:14 (2012), 3224–3255. MR Zbl

[16] M. B. Guillén and D. Maclagan, "A presentation for the Cox ring of $\overline{M}_{0,6}$", preprint, 2017. Zbl arXiv 1712.08193

[17] R. Hartshorne, *Algebraic geometry*, Graduate Texts in Mathematics **52**, Springer, New York-Heidelberg, 1977. MR Zbl

[18] B. Hassett and Y. Tschinkel, "On the effective cone of the moduli space of pointed rational curves", pp. 83–96 in *Topology and geometry*: *commemorating SISTAG*, edited by A. J. Berrick et al., Contemp. Math. **314**, American Mathematical Society, Providence, RI, 2002. MR Zbl

[19] J. Huh, "Milnor numbers of projective hypersurfaces and the chromatic polynomial of graphs", *J. Amer. Math. Soc.* **25**:3 (2012), 907–927. MR Zbl

[20] J. Huh and E. Katz, "Log-concavity of characteristic polynomials and the Bergman fan of matroids", *Math. Ann.* **354**:3 (2012), 1103–1116. MR Zbl

[21] E. Katz, "A tropical toolkit", *Expo. Math.* **27**:1 (2009), 1–36. MR Zbl

[22] E. Katz, "Tropical intersection theory from toric varieties", *Collect. Math.* **63**:1 (2012), 29–44. MR Zbl

[23] E. Katz and S. Payne, "Realization spaces for tropical fans", pp. 73–88 in *Combinatorial aspects of commutative algebra and algebraic geometry*, edited by G. Fløystad et al., Abel Symp. **6**, Springer, Berlin, 2011. MR Zbl

[24] S. Keel, "Intersection theory of moduli space of stable $n$-pointed curves of genus zero", *Trans. Amer. Math. Soc.* **330**:2 (1992), 545–574. MR Zbl

[25] S. Keel and J. McKernan, "Contractible extremal rays on $\overline{M}_{0,n}$", pp. 115–130 in *Handbook of moduli*, *II*, edited by G. Farkas and I. Morrison, Adv. Lect. Math. (ALM) **25**, Int. Press, Somerville, MA, 2013. MR Zbl

[26] D. R. Grayson and M. E. Stillman, "Macaulay2, a software system for research in algebraic geometry", available at https://faculty.math.illinois.edu/Macaulay2/. Zbl

[27] D. Maclagan and B. Sturmfels, *Introduction to tropical geometry*, Graduate Studies in Mathematics **161**, American Mathematical Society, Providence, RI, 2015. MR Zbl

[28] M. Opie, "Extremal divisors on moduli spaces of rational curves with marked points", *Michigan Math. J.* **65**:2 (2016), 251–285. MR Zbl

[29] B. Osserman and S. Payne, "Lifting tropical intersections", *Doc. Math.* **18** (2013), 121–175. MR Zbl

[30] K. M. Shaw, "A tropical intersection product in matroidal fans", *SIAM J. Discrete Math.* **27**:1 (2013), 459–491. MR Zbl

[31] M. D. Sikirić and E. Jovinelly, "Extreme divisors on $\bar{M}_{0,7}$ and differences over characteristic 2", preprint, 2022. Zbl arXiv 2203.13917

[32] R. P. Stanley, "An introduction to hyperplane arrangements", pp. xii+691 in *Geometric combinatorics* (Park City, UT, 2004), edited by E. Miller et al., IAS/Park City Mathematics Series **13**, American Mathematical Society, Providence, RI, 2007. MR Zbl

[33] R. P. Stanley, *Enumerative combinatorics, 1*, 2nd ed., Cambridge Studies in Advanced Mathematics **49**, Cambridge Univ. Press, 2012. MR Zbl

[34] J. Tevelev, "Compactifications of subvarieties of tori", *Amer. J. Math.* **129**:4 (2007), 1087–1104. MR Zbl

[35] P. Vermeire, "A counterexample to Fulton's conjecture on $\overline{M}_{0,n}$", *J. Algebra* **248**:2 (2002), 780–784. MR Zbl

RECEIVED: 27 Sep 2022 REVISED: 27 Jul 2023 ACCEPTED: 14 Sep 2023

ALESSIO BORZÌ:
alessio.borzi@mis.mpg.de
Max-Planck-Institut für Mathematik in den Naturwissenschaften, Leipzig, Germany